

**МИНИСТЕРСТВО НАУКИ И ВЫСШЕГО ОБРАЗОВАНИЯ
РОССИЙСКОЙ ФЕДЕРАЦИИ**
Федеральное государственное автономное образовательное учреждение
высшего образования «Севастопольский государственный университет»

Институт информационных технологий и управления в технических
системах
Кафедра корпоративных информационных систем

ПЛАТФОРМА 1С

МЕТОДИЧЕСКИЕ УКАЗАНИЯ
к выполнению лабораторной работы № 6
по дисциплине
для студентов очной формы обучения
по направлениям подготовки:

09.03.01 «Информатика и вычислительная техника»,
09.03.02 «Информационные системы и технологии»,
09.03.03 «Прикладная информатика»,
09.03.04 «Программная инженерия»

В 8 частях
Часть 6

Севастополь
СевГУ
2023

УДК 004.45:65.011.56(076.5)
ББК 32.973-018.2я73
ПЗ76

Рецензент:

Заморенова Д. В. – канд. техн. наук, доцент кафедры ПСАТП

Составители:

Шевченко В. И. – канд. техн. наук, зав. базовой кафедрой КИС,
Виноградов Е. В. – доцент базовой кафедры КИС,
Кудашев В. С. – доцент базовой кафедры КИС,
Сергеев Н. О. – ассистент кафедры КИС

ПЗ36 Платформа 1С. В 8 частях. Часть 6: методические указания для выполнения лабораторной работы №6 по дисциплине для студентов очной формы обучения по направлениям подготовки: 09.03.01 «Информатика и вычислительная техника», 09.03.02 «Информационные системы и технологии», 09.03.03 «Прикладная информатика», 09.03.04 «Программная инженерия» / Севастопольский государственный университет, Институт информационных технологий и управления в технических системах, Базовая кафедра корпоративных информационных систем; сост.: Шевченко В. И., Виноградов Е. В., Кудашев В. С., Сергеев Н. О. – Севастополь: СевГУ, 2023. – **56** с.

Целью методических указаний является оказание помощи студентам в выполнении лабораторных работ по дисциплине. Методические указания содержат общие положения и требования к отчету, основные теоретические сведения, описание вариантов индивидуальных заданий.

УДК **004.45:65.011.56(076.5)**
ББК **32.973-018.2я73**

Методические указания рассмотрены и утверждены на заседании базовой кафедры «Корпоративные информационные системы» (протокол №**10** от **26** мая **2022** г.)

Ответственный за выпуск:

© СевГУ, ИИТиУТС, 2023

© Шевченко В. И., Виноградов Е. В., Кудашев В. С., Сергеев Н. О., 2023

Издательский номер

СОДЕРЖАНИЕ

ЛАБОРАТОРНАЯ РАБОТА №6. СТРАТЕГИИ СПИСАНИЯ ВЗАИМОРАСЧЕТОВ (ПО СРЕДНЕЙ, ФИФО, ЛИФО). УПРАВЛЯЕМЫЕ БЛОКИРОВКИ	4
1. ТЕОРЕТИЧЕСКИЕ СВЕДЕНИЯ	4
1.1 Методы списания взаиморасчетов	4
1.2 Блокировка записей	4
1.3 Одновременное сведение ресурсов к нулю и списание «последней копейки»	5
1.4 Системная транзакция и свойство «Отказ»	6
2. ПРИМЕР ВЫПОЛНЕНИЯ РАБОТЫ	7
2.1 Задание	7
2.2 Ход работы	7
ЗАДАНИЕ НА ЛАБОРАТОРНУЮ РАБОТУ	11
СОДЕРЖАНИЕ ОТЧЕТА	23
КОНТРОЛЬНЫЕ ВОПРОСЫ	24

ЛАБОРАТОРНАЯ РАБОТА №6. СТРАТЕГИИ СПИСАНИЯ ВЗАИМОРАСЧЕТОВ (ПО СРЕДНЕЙ, ФИФО, ЛИФО). УПРАВЛЯЕМЫЕ БЛОКИРОВКИ

Цель работы:

Изучить и научиться: порядок списания взаиморасчетов, принцип управляемых блокировок.

Используемое программное обеспечение: учебная версия платформы «1С: Предприятие 8.3».

Режим совместимости: «Версия 8.3.9».

1. ТЕОРЕТИЧЕСКИЕ СВЕДЕНИЯ

1.1 Методы списания взаиморасчетов

Важной особенностью списания долгов, себестоимости и тому подобное является строгое правило, устанавливающее порядок списания. Самый простой метод списания – это стратегия «По средней». Берем всю оставшуюся сумму, делим на оставшееся количество – и готово! Просто, но не очень, потому что себестоимость при изменчивости цен будет какой-то непонятной типа «123,45», а не «125,00». Потому что в разных партиях цена будет разной. Поэтому применяется списание по партиям (документам взаиморасчетов). Но надо установить порядок. Основными стратегиями списания по документам взаиморасчетов являются:

- ФИФО (от англ. «FIFO» – First In, First Out) – первым списывается тот, кто раньше был оформлен.

- ЛИФО (от англ. «LIFO» – Last In, First Out) – последним списывается тот, кто раньше был оформлен.

Законодательно утвержден порядок «ФИФО». Но это – в регламентированном учете. А в управленческом – на усмотрение владельца бизнеса.

Аналогия с методами списания по документам взаиморасчетов – это «Стек» (ЛИФО) и «Очередь» (ФИФО).

1.2 Блокировка записей

Управляемые блокировки используются при работе сразу нескольких пользователей в одной информационной базе. Например, если двое пользователей пытаются продать оставшиеся остатки телефонов, то без блокировки этих остатков будут по учету два списания. И на остатках останется отрицательное количество телефонов. В реальности же будет драка разбор случившейся ситуации.

Виды блокировок бывают разные. В 1С:Предприятие есть два вида блокировок.

Первая блокировка – разделяемая. Это означает, что один пользователь блокирует запись, не разрешая записывать в эту запись, но при этом разрешает прочитать данные из этой записи. Пример: я могу прочитать, сколько осталось телефонов только для индикации этого количества, а не для продажи.

Вторая блокировка – эксклюзивная. Пользователь, установивший такую блокировку, не разрешает ни читать данные заблокированной записи, ни писать в эту запись. Пример: пользователь уже собирается продавать остатки телефонов, поэтому ни показывать, ни тем более продавать мне нельзя.

Мы имеем доступ к физической таблице движений регистра накопления. А вот к физической таблице итогов мы доступа не ~~достойны~~ имеем. Как же заблокировать остатки (и обороты)? А очень легко: остатки получаются из движений и итогов (лабораторная работа №5). Поэтому блокировка только движений автоматически блокирует и остатки.

Для избегания простоев на блокировках очень желательно блокировать минимум записей. Объект БлокировкаДанных использует коллекцию элементов блокировки, функционирующих по правилу ИЛИ. В примере для данной лабораторной работы можно было заблокировать остатки товаров по полям Номенклатура и ДокументВзаиморасчетов. Это сильно поможет установить точечные блокировки, но при этом для выяснения списка документов взаиморасчетов придется имитировать списание, потом заблокировать записи с конкретными документами взаиморасчетов, и уж потом выполнить реальное списание. Учитывая, что обычно спектр номенклатур очень широк, список документов практически не влияет на производительность. Поэтому имеет смысл заблокировать ВСЕ записи, но для КОНКРЕТНОЙ номенклатуры.

1.3 Одновременное сведение ресурсов к нулю и списание «последней копейки»

Допустим, на остатках есть 3 единицы товара на сумму 100 ~~тысяч~~ рублей. Списываем одну единицу. Стоимость списания: $100 / 3 = 33,33$ (не в периоде, потому что копейки неделимые). Остаток суммы: $100 - 33,33 = 66,67$. Вторую списываем, на остатке 33,34. Третью списываем, на остатке... 0,01. Казалось бы, ну и что такого, копейка – это ничто! Ну, во-первых, «за ошибки в бухгалтерском учете штрафуют, а за ошибки в налоговом учете сажают». А во-вторых, количество-то ноль, т.е. нечего больше списывать, а копейка будет дублироваться в таблице итогов каждый месяц и во веки веков. База данных будет раздуваться, а производительность будет снижаться. И тут для повышения производительности и увеличения памяти уже не обойдется какими-то копейками.

Как быть? А при последнем списании, когда количество списываемого товара в точности равно количеству на остатках, не надо вычислять сумму списания, а надо списывать всю сумму полностью. Сложно? 😊

1.4 Системная транзакция и свойство «Отказ»

В 1С работа с базой данных обычно происходит в транзакции. Это означает, что после окончания работы есть два варианта: зафиксировать транзакцию или отменить транзакцию. Это решает многие проблемы. Самая важная из них – возможность восстановить состояние базы данных в исходное состояние до начала акта (т.е. транзакции) работы с базой данных. 1С не поддерживает вложенные транзакции.

При записи объектов в базу данных инициирована системная транзакция. Эта транзакция одна на все обработчики в порядке ПередЗаписью, ПриЗаписи, ОбработкаУдаленияПроведения, ОбработкаПроведения. Если любом обработчике будет присваивание свойству «Отказ» значение «Истина», то цепочка обработчиков будет разорвана, системная транзакция будет откатана. При этом отмена транзакции будет выполнена не в момент присваивания, а при завершении обработчика, в котором этом присваивание было осуществлено.

2. ПРИМЕР ВЫПОЛНЕНИЯ РАБОТЫ

2.1 Задание

Предметная область: Оптовая торговля.

Описание деятельности: Склад "Пропать" производит закупку и оптовую продажу товаров (номенклатура). Предприятие работает с поставщиками и покупателями (контрагенты). В любой момент может измениться цена номенклатуры как со стороны склада, так и со стороны поставщика. Продажа осуществляется в соответствии с учетной политикой (стратегии по средней, ФИФО, ЛИФО). Предприятием предоставляется отчетность.

Необходимо сделать:

1. Написать блокировку записей регистра ОстаткиТоваров перед списанием остатков. Желательно (точнее, обязательно) минимизировать количество блокируемых записей.
2. Переписать обработчик проведения документа ПродажаТоваров по регистру ОстаткиТоваров, используя стратегию учета ПоСредней.
3. Переписать обработчик проведения документа ПродажаТоваров по регистру ОстаткиТоваров, используя стратегию учета ФИФО.
4. Переписать обработчик проведения документа ПродажаТоваров по регистру ОстаткиТоваров, используя стратегию учета ЛИФО.

2.2 Ход работы

Напишем блокировку записей, содержащих данные номенклатуры, подлежащей списанию (рисунок 2.1). Этот и следующий код расположен в модуле объекта документа «Продажа товаров».

```
// регистр ОстаткиТоваров Расход
Движения.ОстаткиТоваров.Записывать = Истина;

// Блокировка номенклатуры для списания
ЗапросДляБлокировки = Новый Запрос();
ЗапросДляБлокировки.Текст = "ВЫБРАТЬ
|      ПродажаТоваровТовары.Номенклатура КАК Номенклатура
|ИЗ
|      Документ.ПродажаТоваров.Товары КАК ПродажаТоваровТовары
|ГДЕ
|      ПродажаТоваровТовары.Ссылка = &Ссылка
|
|СГРУППИРОВАТЬ ПО
|      ПродажаТоваровТовары.Номенклатура";
ЗапросДляБлокировки.УстановитьПараметр("Ссылка", Ссылка);

Блокировка = Новый БлокировкаДанных();
ЭлементБлокировки = Блокировка.Добавить("РегистрНакопления.ОстаткиТоваров");
ЭлементБлокировки.ИсточникДанных = ЗапросДляБлокировки.Выполнить();
ЭлементБлокировки.ИспользоватьИзИсточникаДанных("Номенклатура", "Номенклатура");
ЭлементБлокировки.Режим = РежимБлокировкиДанных.Исключительный;
Блокировка.Заблокировать();
```

Рисунок 2.1 – Блокировка записей

Для стратегии списания «По средней» код представлен на рисунке 2.2.
 Для стратегий списания «ФИФО» и «ЛИФО» код представлен на рисунке 2.3.
 Результаты применения стратегий показаны на рисунке 2.4.

```

СтратегияСписания = РегистрыСведений.УчетнаяПолитика.ПолучитьПоследнее(Дата).СтратегияУчета;

Если СтратегияСписания = Перечисления.СтратегииУчета.ПоСредней ИЛИ
СтратегияСписания = Перечисления.СтратегииУчета.ПустаяСсылка() Тогда
    Запрос = Новый Запрос();
    Запрос.Текст = "ВЫБРАТЬ
        | ПродажаТоваровТовары.Номенклатура КАК Номенклатура,
        | СУММА(ПродажаТоваровТовары.Количество) КАК Количество
        | ПОМЕСТИТЬ ВТТовары
        | ИЗ
        | Документ.ПродажаТоваров.Товары КАК ПродажаТоваровТовары
        | ГДЕ
        | ПродажаТоваровТовары.Ссылка = &Ссылка
        |
        | СГРУППИРОВАТЬ ПО
        | ПродажаТоваровТовары.Номенклатура
        | ;
        |
        | //////////////////////////////////////
        | ВЫБРАТЬ
        | ВТТовары.Номенклатура КАК Номенклатура,
        | СУММА(ЕСТЬNULL(ОстаткиТоваровОстатки.КоличествоОстаток, 0)) КАК КоличествоОстаток,
        | МАКСИМУМ(ВТТовары.Количество) КАК Количество,
        | СУММА(ЕСТЬNULL(ОстаткиТоваровОстатки.СуммаОстаток, 0)) КАК СуммаОстаток
        | ИЗ
        | ВТТовары КАК ВТТовары
        | ЛЕВОЕ СОЕДИНЕНИЕ РегистрНакопления.ОстаткиТоваров.Остатки(
        |     &Период,
        |     Номенклатура В
        |     (ВЫБРАТЬ
        |         | ВТТовары.Номенклатура КАК Номенклатура
        |         | ИЗ
        |         | ВТТовары КАК ВТТовары)) КАК ОстаткиТоваровОстатки
        | ПО ВТТовары.Номенклатура = ОстаткиТоваровОстатки.Номенклатура
        |
        | СГРУППИРОВАТЬ ПО
        | ВТТовары.Номенклатура";
    Запрос.УстановитьПараметр("Период", Новый Граница(МоментВремени(), ВидГраницы.Исключая));
    Запрос.УстановитьПараметр("Ссылка", Ссылка);
    Выборка = Запрос.Выполнить().Выбрать();
    Пока Выборка.Следующий() Цикл

        Если Выборка.Количество > Выборка.КоличествоОстаток Тогда
            Сообщить("Товар: <" + Выборка.Номенклатура + ">. Нехватка: " +
                (Выборка.Количество - Выборка.КоличествоОстаток));
            Отказ = Истина;
            Продолжить;
        КонецЕсли;

        Если Выборка.Количество = Выборка.КоличествоОстаток Тогда
            // списание "последней копейки"
            СуммаСписания = Выборка.СуммаОстаток;
        Иначе
            СуммаСписания = Выборка.Количество * (Выборка.СуммаОстаток / Выборка.КоличествоОстаток);
        КонецЕсли;

        Движение = Движения.ОстаткиТоваров.Добавить();
        Движение.ВидДвижения = ВидДвиженияНакопления.Расход;
        Движение.Период = Дата;
        Движение.Номенклатура = Выборка.Номенклатура;
        Движение.ДокументВзаиморасчетов = Документы.ПокупкаТоваров.ПустаяСсылка();
        Движение.Количество = Выборка.Количество;
        Движение.Сумма = СуммаСписания;

    КонецЦикла;

    Если Отказ Тогда
        Возврат;
    КонецЕсли;

```

Рисунок 2.2 – Списание по стратегии «По средней»

```

ИначеЕсли СтратегияСписания = Перечисления.СтратегииУчета.ФИФО ИЛИ
СтратегияСписания = Перечисления.СтратегииУчета.ЛИФО Тогда

    УсловиеСортировки = "";

Если СтратегияСписания = Перечисления.СтратегииУчета.ЛИФО Тогда
    УсловиеСортировки = " УВЫВ";
КонецЕсли;

Запрос = Новый Запрос();
Запрос.Текст = "ВЫБРАТЬ
|     ПродажаТоваровТовары.Номенклатура КАК Номенклатура,
|     СУММА(ПродажаТоваровТовары.Количество) КАК Количество,
|     СУММА(ПродажаТоваровТовары.Сумма) КАК Сумма
| ПОМЕСТИТЬ ВТТовары
| ИЗ
|     Документ.ПродажаТоваров.Товары КАК ПродажаТоваровТовары
| ГДЕ
|     ПродажаТоваровТовары.Ссылка = &Ссылка
|
| СГРУППИРОВАТЬ ПО
|     ПродажаТоваровТовары.Номенклатура
| ;
|
| //////////////////////////////////////
| ВЫБРАТЬ
|     ОстаткиТоваровОстатки.Номенклатура КАК Номенклатура,
|     ОстаткиТоваровОстатки.ДокументВзаиморасчетов КАК ДокументВзаиморасчетов,
|     ОстаткиТоваровОстатки.КоличествоОстаток КАК КоличествоОстаток,
|     ОстаткиТоваровОстатки.СуммаОстаток КАК СуммаОстаток,
|     ВТТовары.Количество КАК Количество,
|     ВТТовары.Сумма КАК Сумма
| ИЗ
|     ВТТовары КАК ВТТовары
|     ЛЕВОЕ СОЕДИНЕНИЕ РегистрНакопления.ОстаткиТоваров.Остатки (
|         &Период,
|         Номенклатура В
|         (ВЫБРАТЬ
|             ВТТовары.Номенклатура КАК Номенклатура
|             ИЗ
|                 ВТТовары КАК ВТТовары)) КАК ОстаткиТоваровОстатки
|     ПО ВТТовары.Номенклатура = ОстаткиТоваровОстатки.Номенклатура
|
| УПОРЯДОЧИТЬ ПО
|     ОстаткиТоваровОстатки.ДокументВзаиморасчетов.МоментВремени " + УсловиеСортировки + "
| ИТОГИ
|     СУММА(КоличествоОстаток),
|     СУММА(СуммаОстаток),
|     МАКСИМУМ(Количество),
|     МАКСИМУМ(Сумма)
| ПО
|     Номенклатура";

Запрос.УстановитьПараметр("Период", Новый Граница(МоментВремени(), ВидГраницы.Исключая));
Запрос.УстановитьПараметр("Ссылка", Ссылка);

ВыборкаИтоги = Запрос.Выполнить().Выбрать(ОбходРезультатаЗапроса.ПоГруппировкам);
Пока ВыборкаИтоги.Следующий() Цикл
    Если ВыборкаИтоги.Количество > ВыборкаИтоги.КоличествоОстаток Тогда
        Сообщить("Товар: <" + ВыборкаИтоги.Номенклатура + ">. Нехватка: " +
            (ВыборкаИтоги.Количество - ВыборкаИтоги.КоличествоОстаток));
        Отказ = Истина;
        Продолжить;
    КонецЕсли;
    КоличествоСписания = ВыборкаИтоги.Количество;
    ВыборкаДетали = ВыборкаИтоги.Выбрать();
    Пока ВыборкаДетали.Следующий() Цикл
        Если КоличествоСписания = 0 Тогда
            Прервать;
        КонецЕсли;
        Количество = Мин(КоличествоСписания, ВыборкаДетали.КоличествоОстаток);
        Если Количество = ВыборкаДетали.КоличествоОстаток Тогда
            // Списание "последней копейки"
            СуммаСписания = ВыборкаДетали.СуммаОстаток;
        Иначе
            СуммаСписания = Количество * (ВыборкаДетали.СуммаОстаток / ВыборкаДетали.КоличествоОстаток);
        КонецЕсли;

        Движение = Движения.ОстаткиТоваров.Добавить();
        Движение.ВидДвижения = ВидДвиженияНакопления.Расход;
        Движение.Период = Дата;
        Движение.Номенклатура = ВыборкаДетали.Номенклатура;
        Движение.ДокументВзаиморасчетов = ВыборкаДетали.ДокументВзаиморасчетов;
        Движение.Количество = Количество;
        Движение.Сумма = СуммаСписания;

        КоличествоСписания = КоличествоСписания - Количество;
    КонецЦикла;
КонецЦикла;
КонецЕсли;

```

Рисунок 2.3 – Списание по стратегиям «ФИФО» и «ЛИФО»

↓	Регистра...	Номер ...	Номенклатура	Документ взаиморасчетов	Количество	Сумма
33	Покупка ...	1	Телефон		3,000	126 000,00
06	Покупка ...	1	Телефон		3,000	132 000,00
08	Продаж...	1	Телефон		5,000	215 000,00

↓	Регистра...	Номер ...	Номенклатура	Документ взаиморасчетов	Количество	Сумма
33	Покупка ...	1	Телефон	Покупка товаров 000000001...	3,000	126 000,00
06	Покупка ...	1	Телефон	Покупка товаров 000000002...	3,000	132 000,00
08	Продаж...	1	Телефон	Покупка товаров 000000001...	3,000	126 000,00
08	Продаж...	2	Телефон	Покупка товаров 000000002...	2,000	88 000,00

↓	Регистра...	Номер ...	Номенклатура	Документ взаиморасчетов	Количество	Сумма
33	Покупка ...	1	Телефон	Покупка товаров 000000001...	3,000	126 000,00
06	Покупка ...	1	Телефон	Покупка товаров 000000002...	3,000	132 000,00
08	Продаж...	1	Телефон	Покупка товаров 000000002...	3,000	132 000,00
08	Продаж...	2	Телефон	Покупка товаров 000000001...	2,000	84 000,00

Рисунок 2.4 – Результаты применения стратегий «По средней», «ФИФО» и «ЛИФО»

ЗАДАНИЕ НА ЛАБОРАТОРНУЮ РАБОТУ

Необходимо взять конфигурацию как результат предыдущих лабораторных работ и доработать при выполнении текущей лабораторной работы. Номер варианта образуется как остаток от делений номера зачетки на 25, например, если номер зачетки равен 123456, тогда остаток от деления на 25 равен 6, то есть это и есть номер варианта.

Вариант 0

Предметная область: Гостиничный бизнес.

Описание деятельности: Гостиница "Хостел 10 звезд" осуществляет бронирование номеров и прием оплаты от клиентов (контрагенты) на определенное количество гостей (количество). В любой момент может измениться тариф на номер с учетом типа номера. Заккрытие долга осуществляется в соответствии с учетной политикой (стратегии ФИФО, ЛИФО). Компанией предоставляется отчетность.

Необходимо сделать:

1. Написать блокировку записей регистра ОстаткиБрони перед списанием остатков. Желательно (точнее, обязательно) минимизировать количество блокируемых записей.
2. Переписать обработчик проведения документа Оплата по регистру ОстаткиБрони, используя стратегию учета ПоСредней.
3. Переписать обработчик проведения документа Оплата по регистру ОстаткиБрони, используя стратегию учета ФИФО.
4. Переписать обработчик проведения документа Оплата по регистру ОстаткиБрони, используя стратегию учета ЛИФО.

Вариант 1

Предметная область: Туристический бизнес.

Описание деятельности: Турфирма "Путешествуй по всей России!" оформляет горящие путевки на определенное количество туристов (количество) и принимает оплату от клиентов (контрагенты). В любой момент может измениться цена путевки с учетом типа путевки. Заккрытие долга осуществляется в соответствии с учетной политикой (стратегии ФИФО, ЛИФО). Турфирмой предоставляется отчетность.

Необходимо сделать:

1. Написать блокировку записей регистра ОстаткиОформлений перед списанием остатков. Желательно (точнее, обязательно) минимизировать количество блокируемых записей.
2. Переписать обработчик проведения документа Оплата по регистру ОстаткиОформлений, используя стратегию учета ПоСредней.
3. Переписать обработчик проведения документа Оплата по регистру ОстаткиОформлений, используя стратегию учета ФИФО.
4. Переписать обработчик проведения документа Оплата по регистру ОстаткиОформлений, используя стратегию учета ЛИФО.

Вариант 2

Предметная область: Изготовление микроэлектроники.

Описание деятельности: Завод микроэлектроники "Наноэлектроника" изготавливает микросхемы по технологии 5 нанометров и принимает заказы от клиентов на такие чипы. Оплата заказов осуществляется постфактум. В любой момент может измениться цена на чип с учетом типа микросхемы. Заккрытие долга осуществляется в соответствии с учетной политикой (стратегии ФИФО, ЛИФО). Заводом предоставляется отчетность.

Необходимо сделать:

1. Написать блокировку записей регистра ОстаткиЗаказов перед списанием остатков. Желательно (точнее, обязательно) минимизировать количество блокируемых записей.
2. Переписать обработчик проведения документа Оплата по регистру ОстаткиЗаказов, используя стратегию учета ПоСредней.
3. Переписать обработчик проведения документа Оплата по регистру ОстаткиЗаказов, используя стратегию учета ФИФО.
4. Переписать обработчик проведения документа Оплата по регистру ОстаткиЗаказов, используя стратегию учета ЛИФО.

Вариант 3

Предметная область: Ремонт судов.

Описание деятельности: Судоремонтный завод "Морской бой" выполняет ремонт кораблей для компаний-судовладельцев (контрагенты). Конечно, нужно заплатить за работу после ремонта. В любой момент может измениться стоимость ремонта корабля с учетом типа судна. Заккрытие оплаты за ремонт осуществляется в соответствии с учетной политикой (стратегии ФИФО, ЛИФО). Заводом предоставляется отчетность.

Необходимо сделать:

1. Написать блокировку записей регистра ОстаткиРемонтов перед списанием остатков. Желательно (точнее, обязательно) минимизировать количество блокируемых записей.
2. Переписать обработчик проведения документа Оплата по регистру ОстаткиРемонтов, используя стратегию учета ПоСредней.
3. Переписать обработчик проведения документа Оплата по регистру ОстаткиРемонтов, используя стратегию учета ФИФО.
4. Переписать обработчик проведения документа Оплата по регистру ОстаткиРемонтов, используя стратегию учета ЛИФО.

Вариант 4

Предметная область: Парикмахерская деятельность.

Описание деятельности: Парикмахерская "Элитная грива" осуществляет предварительный заказы на стрижку и создание причесок с указанием компании (контрагенты) и количества подстригаемых (количество). После выполнения работы нужно парикмахерской заплатить. В любой момент

может измениться стоимость стрижки/прически с учетом типа услуг. Заккрытие оплаты работы осуществляется в соответствии с учетной политикой (стратегии ФИФО, ЛИФО). Парикмахерской предоставляется отчетность.

Необходимо сделать:

1. Написать блокировку записей регистра ОстаткиЗаказов перед списанием остатков. Желательно (точнее, обязательно) минимизировать количество блокируемых записей.
2. Переписать обработчик проведения документа Оплата по регистру ОстаткиЗаказов, используя стратегию учета ПоСредней.
3. Переписать обработчик проведения документа Оплата по регистру ОстаткиЗаказов, используя стратегию учета ФИФО.
4. Переписать обработчик проведения документа Оплата по регистру ОстаткиЗаказов, используя стратегию учета ЛИФО.

Вариант 5

Предметная область: Адвокатская деятельность.

Описание деятельности: Адвокатское бюро "Броня и щит" занимается защитой состоятельных клиентов (контрагенты) с определенным количеством дел (количество) и принимает ~~взятки~~ оплату оказанных услуг. В любой момент может измениться стоимость адвокатских услуг с учетом типа услуг. Заккрытие оплаты услуг осуществляется в соответствии с учетной политикой (стратегии ФИФО, ЛИФО). Бюро предоставляется отчетность.

Необходимо сделать:

1. Написать блокировку записей регистра ОстаткиЗащит перед списанием остатков. Желательно (точнее, обязательно) минимизировать количество блокируемых записей.
2. Переписать обработчик проведения документа Оплата по регистру ОстаткиЗащит, используя стратегию учета ПоСредней.
3. Переписать обработчик проведения документа Оплата по регистру ОстаткиЗащит, используя стратегию учета ФИФО.
4. Переписать обработчик проведения документа Оплата по регистру ОстаткиЗащит, используя стратегию учета ЛИФО.

Вариант 6

Предметная область: Торговля валютой.

Описание деятельности: Обменник "Сто долларов и два в уме" два раза подумает перед тем, как принять заказ на покупку валют и потом принять оплату заказа. В каждый момент может измениться курс валюты. Заккрытие долга за заказы осуществляется в соответствии с учетной политикой (стратегии ФИФО, ЛИФО). Обменником предоставляется отчетность.

Необходимо сделать:

1. Написать блокировку записей регистра ОстаткиЗаказов перед списанием остатков. Желательно (точнее, обязательно) минимизировать количество блокируемых записей.

2. Переписать обработчик проведения документа Оплата по регистру ОстаткиЗаказов, используя стратегию учета ПоСредней.

3. Переписать обработчик проведения документа Оплата по регистру ОстаткиЗаказов, используя стратегию учета ФИФО.

4. Переписать обработчик проведения документа Оплата по регистру ОстаткиЗаказов, используя стратегию учета ЛИФО.

Вариант 7

Предметная область: Торговля лекарствами.

Описание деятельности: Аптека "Оптика" закупает льготные лекарства и бесплатно раздает их льготникам. За это получает дотации из бюджета задним числом через посредников (контрагенты). В любой момент может измениться цена лекарств с учетом типа лекарств. Заккрытие оплаты лекарств осуществляется в соответствии с учетной политикой (стратегии ФИФО, ЛИФО). Аптекой предоставляется отчетность.

Необходимо сделать:

1. Написать блокировку записей регистра ОстаткиВыдач перед списанием остатков. Желательно (точнее, обязательно) минимизировать количество блокируемых записей.

2. Переписать обработчик проведения документа Дотация по регистру ОстаткиВыдач, используя стратегию учета ПоСредней.

3. Переписать обработчик проведения документа Дотация по регистру ОстаткиВыдач, используя стратегию учета ФИФО.

4. Переписать обработчик проведения документа Дотация по регистру ОстаткиВыдач, используя стратегию учета ЛИФО.

Вариант 8

Предметная область: Приготовление еды.

Описание деятельности: Кафе "Кафе – это кафе (кафе)" готовит блюда для доставки на дом. После этого клиентом (контрагенты) осуществляется оплата блюд, возможно, порциями. В любой момент может измениться цена блюда с учетом типа блюд. Заккрытие оплаты блюд осуществляется в соответствии с учетной политикой (стратегии ФИФО, ЛИФО). Кафе предоставляется отчетность.

Необходимо сделать:

1. Написать блокировку записей регистра ОстаткиДоставок перед списанием остатков. Желательно (точнее, обязательно) минимизировать количество блокируемых записей.

2. Переписать обработчик проведения документа Оплата по регистру ОстаткиДоставок, используя стратегию учета ПоСредней.

3. Переписать обработчик проведения документа Оплата по регистру ОстаткиДоставок, используя стратегию учета ФИФО.

4. Переписать обработчик проведения документа Оплата по регистру ОстаткиДоставок, используя стратегию учета ЛИФО.

Вариант 9

Предметная область: Розыскное дело.

Описание деятельности: Детективное бюро "Нюх века" следит за всеми, но за денежку. Готово работать в долг, принимая заказ на слежку в рамках нескольких дел (количество). В любой момент может измениться стоимость услуг с учетом типа услуг. Заккрытие оплаты услуг осуществляется в соответствии с учетной политикой (стратегии ФИФО, ЛИФО). Бюро предоставляется отчетность.

Необходимо сделать:

1. Написать блокировку записей регистра ОстаткиСлежек перед списанием остатков. Желательно (точнее, обязательно) минимизировать количество блокируемых записей.
2. Переписать обработчик проведения документа Оплата по регистру ОстаткиСлежек, используя стратегию учета ПоСредней.
3. Переписать обработчик проведения документа Оплата по регистру ОстаткиСлежек, используя стратегию учета ФИФО.
4. Переписать обработчик проведения документа Оплата по регистру ОстаткиСлежек, используя стратегию учета ЛИФО.

Вариант 10

Предметная область: Оказание услуг.

Описание деятельности: Бюро добрых дел "Добродел" ходит за вами и делает добро, правда, не бесплатно. В этом случае принимается заказ на добрые услуги, а после – оплату оказанных добрых услуг. В любой момент может измениться стоимость услуг с учетом типа услуг. Заккрытие оплаты за услуги осуществляется в соответствии с учетной политикой (стратегии ФИФО, ЛИФО). Бюро предоставляется отчетность.

Необходимо сделать:

1. Написать блокировку записей регистра ОстаткиДобрыхУслуг перед списанием остатков. Желательно (точнее, обязательно) минимизировать количество блокируемых записей.
2. Переписать обработчик проведения документа Оплата по регистру ОстаткиДобрыхУслуг, используя стратегию учета ПоСредней.
3. Переписать обработчик проведения документа Оплата по регистру ОстаткиДобрыхУслуг, используя стратегию учета ФИФО.
4. Переписать обработчик проведения документа Оплата по регистру ОстаткиДобрыхУслуг, используя стратегию учета ЛИФО.

Вариант 11

Предметная область: Выполнение переводов.

Описание деятельности: Бюро переводов "Hochesh perevedu?" разбирается в китайской грамоте, и поэтому осуществляет перевод с английского и на хинди. В этом случае принимается заказ на перевод, а после – оплату выполненных работ. В любой момент может измениться цена перевода с учетом языка. Заккрытие оплаты работ осуществляется в соответствии с

учетной политикой (стратегии ФИФО, ЛИФО). Бюро предоставляется отчетность.

Необходимо сделать:

1. Написать блокировку записей регистра ОстаткиТочныхПереводов перед списанием остатков. Желательно (точнее, обязательно) минимизировать количество блокируемых записей.

2. Переписать обработчик проведения документа Оплата по регистру ОстаткиТочныхПереводов, используя стратегию учета ПоСредней.

3. Переписать обработчик проведения документа Оплата по регистру ОстаткиТочныхПереводов, используя стратегию учета ФИФО.

4. Переписать обработчик проведения документа Оплата по регистру ОстаткиТочныхПереводов, используя стратегию учета ЛИФО.

Вариант 12

Предметная область: Торговля на бирже.

Описание деятельности: Брокерская контора "Разгоняй депозит!" шарит в торговле на бирже. Поэтому принимает ордера на покупку/продажу валютных пар от игроков (контрагенты). Потом надо заплатить за ордер. В любой момент может измениться цена валютной пара с учетом типа ордера. Заккрытие оплаты ордеров осуществляется в соответствии с учетной политикой (стратегии ФИФО, ЛИФО). Конторой предоставляется отчетность.

Необходимо сделать:

1. Написать блокировку записей регистра ОстаткиОрдеров перед списанием остатков. Желательно (точнее, обязательно) минимизировать количество блокируемых записей.

2. Переписать обработчик проведения документа Оплата по регистру ОстаткиОрдеров, используя стратегию учета ПоСредней.

3. Переписать обработчик проведения документа Оплата по регистру ОстаткиОрдеров, используя стратегию учета ФИФО.

4. Переписать обработчик проведения документа Оплата по регистру ОстаткиОрдеров, используя стратегию учета ЛИФО.

Вариант 13

Предметная область: Репетиторство.

Описание деятельности: Школа репетиторов "1С – наше все!" знает, что 1С распространен в России чуть больше, чем везде. Поэтому проводит курсы по 1С, формируя группы и принимая оплату. В любой момент может измениться цена курсов с учетом типа курсов. Заккрытие оплаты курсов осуществляется в соответствии с учетной политикой (стратегии ФИФО, ЛИФО). Школой предоставляется отчетность.

Необходимо сделать:

1. Написать блокировку записей регистра ОстаткиКрутыхКурсов перед списанием остатков. Желательно (точнее, обязательно) минимизировать количество блокируемых записей.

2. Переписать обработчик проведения документа Оплата по регистру ОстаткиКрутыхКурсов, используя стратегию учета ПоСредней.

3. Переписать обработчик проведения документа Оплата по регистру ОстаткиКрутыхКурсов, используя стратегию учета ФИФО.

4. Переписать обработчик проведения документа Оплата по регистру ОстаткиКрутыхКурсов, используя стратегию учета ЛИФО.

Вариант 14

Предметная область: Шоу-бизнес.

Описание деятельности: Студия звукозаписи "Семь с половиной струн" поставила на поток производства фонограмм. Богатым певцам нужно заказать фонограмму, а уж после за фонограмму заплатить. В любой момент может измениться стоимость фонограммы с учетом типа фонограмм. Заккрытие оплаты производства фонограмм осуществляется в соответствии с учетной политикой (стратегии ФИФО, ЛИФО). Студией предоставляется отчетность.

Необходимо сделать:

1. Написать блокировку записей регистра ОстаткиПроизводств перед списанием остатков. Желательно (точнее, обязательно) минимизировать количество блокируемых записей.

2. Переписать обработчик проведения документа Оплата по регистру ОстаткиПроизводств, используя стратегию учета ПоСредней.

3. Переписать обработчик проведения документа Оплата по регистру ОстаткиПроизводств, используя стратегию учета ФИФО.

4. Переписать обработчик проведения документа Оплата по регистру ОстаткиПроизводств, используя стратегию учета ЛИФО.

Вариант 15

Предметная область: Актерское дело.

Описание деятельности: Киностудия "Оскаром по голове" начисляет и выплачивает известным и не очень актерам гонорары за съемку фильмов. В любой момент может измениться ставка за съемку в фильме. Выплата гонорара осуществляется в соответствии с учетной политикой (стратегии ФИФО, ЛИФО). Киностудией предоставляется отчетность.

Необходимо сделать:

1. Написать блокировку записей регистра ОстаткиНачисленийГонораров перед списанием остатков. Желательно (точнее, обязательно) минимизировать количество блокируемых записей.

2. Переписать обработчик проведения документа ВыплатаГонораров по регистру ОстаткиНачисленийГонораров, используя стратегию учета ПоСредней.

3. Переписать обработчик проведения документа ВыплатаГонораров по регистру ОстаткиНачисленийГонораров, используя стратегию учета ФИФО.

4. Переписать обработчик проведения документа ВыплатаГонораров по регистру ОстаткиНачисленийГонораров, используя стратегию учета ЛИФО.

Вариант 16

Предметная область: Индивидуальное предпринимательство.

Описание деятельности: Индивидуальный предприниматель Одинокоев Один Единович индивидуально принимает заказ на индивидуальную работу по написанию копий известных художников. После выполнения индивидуальной работы Одинокоеву нужно за индивидуальную работу индивидуально заплатить. В любой момент может измениться стоимость копии картины с учетом типа картин. Закрывание оплаты написания копии картины осуществляется в соответствии с учетной политикой (стратегии ФИФО, ЛИФО). Одинокоевым предоставляется отчетность.

Необходимо сделать:

1. Написать блокировку записей регистра ОстаткиПроизводств перед списанием остатков. Желательно (точнее, обязательно) минимизировать количество блокируемых записей.
2. Переписать обработчик проведения документа Оплата по регистру ОстаткиПроизводств, используя стратегию учета ПоСредней.
3. Переписать обработчик проведения документа Оплата по регистру ОстаткиПроизводств, используя стратегию учета ФИФО.
4. Переписать обработчик проведения документа Оплата по регистру ОстаткиПроизводств, используя стратегию учета ЛИФО.

Вариант 17

Предметная область: Программирование.

Описание деятельности: Отдел разработки "Не разработаешь – отделаю!" ввел штрафы за нарушение сроков отдельных этапов разработки. Эти нарушения фиксируются, а потом виновные во всем не люди платят штраф. В любой момент может измениться ставка штрафа с учетом типа нарушений. Закрывание штрафа за нарушение осуществляется в соответствии с учетной политикой (стратегии ФИФО, ЛИФО). Отделом предоставляется отчетность.

Необходимо сделать:

1. Написать блокировку записей регистра ОстаткиШтрафов перед списанием остатков. Желательно (точнее, обязательно) минимизировать количество блокируемых записей.
2. Переписать обработчик проведения документа ОплатаШтрафа по регистру ОстаткиШтрафов, используя стратегию учета ПоСредней.
3. Переписать обработчик проведения документа ОплатаШтрафа по регистру ОстаткиШтрафов, используя стратегию учета ФИФО.
4. Переписать обработчик проведения документа ОплатаШтрафа по регистру ОстаткиШтрафов, используя стратегию учета ЛИФО.

Вариант 18

Предметная область: Ресторанное дело.

Описание деятельности: Ресторан "Мой сон – гарсон" принимает заказы на студенческую вечеринку. В качестве платы за вечеринку принимаются стипендии и карманные деньги от родителей. В любой момент может измениться тариф за студента с учетом типа вечеринки. Заккрытие оплаты вечеринки осуществляется в соответствии с учетной политикой (стратегии ФИФО, ЛИФО). Рестораном предоставляется отчетность.

Необходимо сделать:

1. Написать блокировку записей регистра ОстаткиВечеринок перед списанием остатков. Желательно (точнее, обязательно) минимизировать количество блокируемых записей.
2. Переписать обработчик проведения документа Оплата по регистру ОстаткиВечеринок, используя стратегию учета ПоСредней.
3. Переписать обработчик проведения документа Оплата по регистру ОстаткиВечеринок, используя стратегию учета ФИФО.
4. Переписать обработчик проведения документа Оплата по регистру ОстаткиВечеринок, используя стратегию учета ЛИФО.

Вариант 19

Предметная область: Транспортные перевозки.

Описание деятельности: АТП "Мощные муравьи" накачало колеса и с помощью дальнобойщиков принялось за прием заказов на перевозки грузов. Понятное дело, что за заказы нужно платить. В любой момент может измениться тариф за перевозку с учетом типа перевозки. Заккрытие оплаты перевозок осуществляется в соответствии с учетной политикой (стратегии ФИФО, ЛИФО). Транспортниками предоставляется отчетность.

Необходимо сделать:

1. Написать блокировку записей регистра ОстаткиЗаказов перед списанием остатков. Желательно (точнее, обязательно) минимизировать количество блокируемых записей.
2. Переписать обработчик проведения документа Оплата по регистру ОстаткиЗаказов, используя стратегию учета ПоСредней.
3. Переписать обработчик проведения документа Оплата по регистру ОстаткиЗаказов, используя стратегию учета ФИФО.
4. Переписать обработчик проведения документа Оплата по регистру ОстаткиЗаказов, используя стратегию учета ЛИФО.

Вариант 20

Предметная область: Заправка топливом.

Описание деятельности: АЗС "Утоли жажду!" жаждет очереди за бензином и утоляет желание сверхприбылей поступившей выручкой. В любой момент может измениться цена за бензин и тип бензина. Заккрытие оплаты бензина осуществляется в соответствии с учетной политикой (стратегии ФИФО, ЛИФО). Заправщиками предоставляется отчетность.

Необходимо сделать:

1. Написать блокировку записей регистра ОстаткиЗаправок перед списанием остатков. Желательно (точнее, обязательно) минимизировать количество блокируемых записей.

2. Переписать обработчик проведения документа Оплата по регистру ОстаткиЗаправок, используя стратегию учета ПоСредней.

3. Переписать обработчик проведения документа Оплата по регистру ОстаткиЗаправок, используя стратегию учета ФИФО.

4. Переписать обработчик проведения документа Оплата по регистру ОстаткиЗаправок, используя стратегию учета ЛИФО.

Вариант 21

Предметная область: Автомобили.

Описание деятельности: Служба такси "Я быстренько" быстренько обслуживает заказы на перевозку пассажиров. Потом служба принимают оплату заказов. В любой момент может измениться тариф за поездку и тип поездки. Заккрытие оплаты поездки осуществляется в соответствии с учетной политикой (стратегии ФИФО, ЛИФО). Службой предоставляется отчетность.

Необходимо сделать:

1. Написать блокировку записей регистра ОстаткиЗаказов перед списанием остатков. Желательно (точнее, обязательно) минимизировать количество блокируемых записей.

2. Переписать обработчик проведения документа Оплата по регистру ОстаткиЗаказов, используя стратегию учета ПоСредней.

3. Переписать обработчик проведения документа Оплата по регистру ОстаткиЗаказов, используя стратегию учета ФИФО.

4. Переписать обработчик проведения документа Оплата по регистру ОстаткиЗаказов, используя стратегию учета ЛИФО.

Вариант 22

Предметная область: Пассажирские перевозки.

Описание деятельности: Трамвайное депо "Конькобежец" бегом принимает заявки на проведение выпускных в трамвае и ждет оплаты выпускных. В любой момент может измениться цена за выпускной и тип выпускного. Заккрытие оплаты выпускного осуществляется в соответствии с учетной политикой (стратегии ФИФО, ЛИФО). Депо предоставляется отчетность.

Необходимо сделать:

1. Написать блокировку записей регистра ОстаткиЗаявок перед списанием остатков. Желательно (точнее, обязательно) минимизировать количество блокируемых записей.

2. Переписать обработчик проведения документа Оплата по регистру ОстаткиЗаявок, используя стратегию учета ПоСредней.

3. Переписать обработчик проведения документа Оплата по регистру ОстаткиЗаявок, используя стратегию учета ФИФО.

4. Переписать обработчик проведения документа Оплата по регистру ОстаткиЗаявок, используя стратегию учета ЛИФО.

Вариант 23

Предметная область: Морские перевозки.

Описание деятельности: Круизный лайнер "Прожарка на море" жарко принимает заявки на туры по морю. После тура путешественникам нужно погасить задолженность. В любой момент может измениться цена за тур и тип тура. Заккрытие оплаты тура осуществляется в соответствии с учетной политикой (стратегии ФИФО, ЛИФО). Лайнером предоставляется отчетность.

Необходимо сделать:

1. Написать блокировку записей регистра ОстаткиЗаявок перед списанием остатков. Желательно (точнее, обязательно) минимизировать количество блокируемых записей.
2. Переписать обработчик проведения документа Оплата по регистру ОстаткиЗаявок, используя стратегию учета ПоСредней.
3. Переписать обработчик проведения документа Оплата по регистру ОстаткиЗаявок, используя стратегию учета ФИФО.
4. Переписать обработчик проведения документа Оплата по регистру ОстаткиЗаявок, используя стратегию учета ЛИФО.

Вариант 24

Предметная область: Мультипликация.

Описание деятельности: Студия анимации "Севастополь, быстрее!" прямо в Севастополе снимает мультипликационные фильмы, причем быстро. Студия принимает заявки на производство мультфильмов и ожидает оплаты этих мультфильмов. В любой момент может измениться ставка за мультфильм и тип мультфильма. Заккрытие оплаты фильма осуществляется в соответствии с учетной политикой (стратегии ФИФО, ЛИФО). Студией предоставляется отчетность.

Необходимо сделать:

1. Написать блокировку записей регистра ОстаткиЗаявок перед списанием остатков. Желательно (точнее, обязательно) минимизировать количество блокируемых записей.
2. Переписать обработчик проведения документа Оплата по регистру ОстаткиЗаявок, используя стратегию учета ПоСредней.
3. Переписать обработчик проведения документа Оплата по регистру ОстаткиЗаявок, используя стратегию учета ФИФО.
4. Переписать обработчик проведения документа Оплата по регистру ОстаткиЗаявок, используя стратегию учета ЛИФО.

СОДЕРЖАНИЕ ОТЧЕТА

Отчет должен содержать:

1. Цель работы.
2. Описание предметной области.
3. Скриншоты с результатами работы в режиме «Предприятие» и с кодом и метаданными в режиме «Конфигуратор».
4. Выводы.

Кроме того, необходимо приложить результаты лабораторной работы в виде dt-файла, содержащего информационную базу.

Подсмотреть, как выполнена лабораторная работа №6 варианта «Эталон», можно в репозитории на гитлабе «1С-Рарус» по адресу: http://gitlab.rarus.ru/что-то_там.dt.

КОНТРОЛЬНЫЕ ВОПРОСЫ

1. Зачем существуют стратегии списания взаиморасчетов?
2. Какая структура данных является аналогией со стратегией «ФИФО»?
3. Какая структура данных является аналогией со стратегией «ЛИФО»?
4. Какие есть виды управляемых блокировок?
5. В чем суть разделяемой блокировки?
6. В чем суть эксклюзивной блокировки?
7. Почему нужно одновременно сводить ресурсы к нулю?
8. В чем особенность работы в базе данных при помощи транзакций?
9. Какие обработчики охватывает системная транзакция?
10. В чем суть свойства «Отказ»?
11. В какой момент происходит откат системной транзакции?
12. Как определить номер вашего варианта лабораторной работы?